

Strongly typed term representations in Coq

Nick Benton · Chung-Kil Hur ·
Andrew Kennedy · Conor McBride

Received: date / Accepted: date

Abstract There are two approaches to formalizing the syntax of typed object languages in a proof assistant or programming language. The *extrinsic* approach is to first define a type that encodes untyped object expressions and then make a separate definition of typing judgements over the untyped terms. The *intrinsic* approach is to make a single definition that captures well-typed object expressions, so ill-typed expressions cannot even be expressed. Intrinsic encodings are attractive and naturally enforce the requirement that metalanguage operations on object expressions, such as substitution, respect object types. The price is that the metalanguage types of intrinsic encodings and operations involve non-trivial dependency, adding significant complexity.

This paper describes intrinsic-style formalizations of both simply-typed and polymorphic languages, and basic syntactic operations thereon, in the Coq proof assistant. The Coq types encoding object-level variables (de Bruijn indices) and terms are indexed by both type and typing environment. One key construction is the boot-strapping of definitions and lemmas about the action of substitutions in terms of similar ones for a simpler notion of renamings. In

N. Benton
Microsoft Research, 7 J J Thomson Avenue, Cambridge, CB3 0FB, United Kingdom
E-mail: nick@example.com

A. J. Kennedy
Microsoft Research, 7 J J Thomson Avenue, Cambridge, CB3 0FB, United Kingdom
E-mail: akenn@microsoft.com

C. McBride
Department of Computer and Information Sciences, University of Strathclyde, 26 Richmond Street, Glasgow, G1 1XH, Scotland
E-mail: conor@strictlypositive.org

C.-K. Hur
Computer Laboratory, University of Cambridge, William Gates Building, 15 J J Thomson Avenue, Cambridge, CB3 0FD, United Kingdom
E-mail: ckh25@cam.ac.uk

the simply-typed case, this yields definitions that are free of any use of type equality coercions. In the polymorphic case, some substitution operations do still require type coercions, which we at least partially tame by uniform use of heterogeneous equality.

Keywords Coq · de Bruijn · strongly-typed

1 Introduction

When encoding a typed object language in a proof assistant or programming language it is common to first define a datatype representing the abstract syntax of object-level expressions, and then make a separate inductive definition of typing judgements, relating expressions to types and type environments. This approach is sometimes referred to as *extrinsic*, and informally equated with Curry’s view of types as *a posteriori* specifications concerning the form or behaviour of the raw untyped expressions. If, as is often the case, one is really only interested in statements about well-typed terms, an attractive alternative is the more Church-style¹ *intrinsic* approach, which builds the type rules into the definition of the abstract syntax right from the start, so all terms are well-typed ‘by construction’. Intrinsic encodings reflect object-level types in metalanguage types, and can naturally and compactly enforce the requirement that operations over object language terms, such as substitution, should respect object-level types. In the extrinsic approach, definitions and lemmas become hedged with extra preconditions that not only add clutter but, when one works with them, have to be repeatedly and explicitly fulfilled – work that can largely be done by the metalanguage type system in an intrinsic encoding.

The price for working with intrinsic encodings is that they require more sophisticated metalanguage types to express the stronger invariants that one is now enforcing on object language expressions. In the functional programming community, ‘strongly typed’ encodings are a popular motivating example for generalized algebraic datatypes (GADTs) (Sheard, 2004; Xi et al, 2003; Cheney and Hinze, 2003), and similar techniques can also be applied in modern object oriented languages (Kennedy and Russo, 2005). In the type theory and automated reasoning community, such ‘internal’ representations have been described by a number of authors (Altenkirch and Reus, 1999; Adams, 2006; McBride, 2005; Danielsson, 2007; Coquand, 2002; McBride and McKinnon, 2004; Harper et al, 1993). So the idea is well-known, and it is intuitively clear that dependently-typed calculi such as CiC have the power to express intrinsic encodings of, for example, simply typed languages. But in practice, however, the pain of actually working with ‘very’ dependent types in systems like Coq seems to have led most programming language researchers to use extrinsic encodings in their mechanized formalizations. A draft by Sozeau

¹ The identification of the two encoding styles with the Church/Curry positions is rather imprecise. In particular, it is common to work with extrinsic contextual typing judgements over a Church-style datatype for syntax, in which binders are annotated with types.

(2007a) describes an intrinsic treatment of simple types and substitutions, but requires many awkward equality coercions.

This paper is a tutorial account of one way of working with intrinsic encodings in Coq, intended to show programming language metatheorists that this really is a viable option, rather than at telling hard-core type-theorists something new. We make no great claims of originality: most of the basic ideas are drawn from papers by Goguen and McKinna (1997), Altenkirch and Reus (1999) and Adams (2006), and an earlier draft on representing simply typed terms and substitutions in Epigram by the fourth author (McBride, 2005). However, we give what we believe to be the first intrinsic treatment of simply-typed terms and substitutions in Coq that is entirely free of coercions. We show how some features added in Coq 8.2 by Sozeau, notably the **dependent destruction** and **Program** tactics, help when working with non-trivial dependencies. We show how the technique extends to richer simply-typed constructs, such as pattern matching, and then apply it to give an intrinsic encoding of a second-order polymorphic language.

The encodings presented here have been successfully used in Coq formalizations of non-trivial results about programming languages. We have formalized the domain-theoretic semantics of the simply-typed language and proved its soundness and adequacy (Benton et al, 2009), and subsequently used that semantics to formulate and prove results on compiler correctness (Benton and Hur, 2009a). The encoding of the polymorphic language has also been used as the basis of a formalization of compiler correctness (Benton and Hur, 2009b).

The Coq code described here is available from the authors' web pages.

2 A simply-typed language

We start with a small simply-typed language with a base type of natural numbers. For presentational purposes, we omit general recursion as this will let us present a simple set-theoretic denotational semantics later on.

In the first part of the paper, we will present the *complete* Coq code for the simply-typed language, albeit not in a strict lexically-scoped order. The following prologue sets up the options and imports from the Coq library that we'll use:

```
Require Import List.
Require Import Program.
Require Import FunctionalExtensionality.
Set Implicit Arguments.
```

The Coq type `Ty` defines the types of the object language, with base type `NAT` and arrow type constructor `ARR`. Since we are using de Bruijn indices, an object level type environment is encoded as a list of types:

```
Inductive Ty := NAT | ARR (ty1 ty2 : Ty).
Definition Env := list Ty.
```

Figure 1 presents typing rules for our (entirely standard) object language in the conventional way, using named binders.

$$\begin{array}{c}
\text{VAR} \frac{x : t \in E}{E \vdash x : t} \quad \text{CONST} \frac{}{E \vdash n : \text{nat}} \quad \text{SUCC} \frac{E \vdash e : \text{nat}}{E \vdash \text{succ } e : \text{nat}} \\
\\
\text{PRED} \frac{E \vdash e : \text{nat}}{E \vdash \text{pred } e : \text{nat}} \quad \text{IFZ} \frac{E \vdash e : \text{nat} \quad E \vdash e_1 : t \quad E \vdash e_2 : t}{E \vdash \text{ifz } e \ e_1 \ e_2 : t} \\
\\
\text{LAM} \frac{E, x : t_1 \vdash e : t_2}{E \vdash \lambda x. e : t_1 \rightarrow t_2} \quad \text{APP} \frac{E \vdash e_1 : t_1 \rightarrow t_2 \quad E \vdash e_2 : t_1}{e_1 \ e_2 : t_2}
\end{array}$$

Fig. 1 The simply-typed lambda calculus with naturals

An *extrinsic* approach to encoding this language would be to define an abstract syntax for expressions, using one's favourite method for representing binding, and then separately to give a typing relation over that syntax. Instead, we take an *intrinsic* approach, combining syntax and typing, by *indexing* the types of variables **Var** and expressions **Exp** by the environment **E** and type **t** for which they are well-typed:

```

Inductive Var : Env → Ty → Type :=
| ZVAR : ∀ E t, Var (t :: E) t
| SVAR : ∀ E t t', Var E t → Var (t' :: E) t.

```

```

Inductive Exp E : Ty → Type :=
| VAR : ∀ t, Var E t → Exp E t
| CONST : nat → Exp E NAT
| SUCC : Exp E NAT → Exp E NAT
| PRED : Exp E NAT → Exp E NAT
| IFZ : ∀ t, Exp E NAT → Exp E t → Exp E t → Exp E t
| LAM : ∀ t1 t2, Exp (t1 :: E) t2 → Exp E (ARR t1 t2)
| APP : ∀ t1 t2, Exp E (ARR t1 t2) → Exp E t1 → Exp E t2.

```

An element of **Var E t** is essentially a derivation establishing that the type **t** is at some position in the list **E**. We count from the left, so the constructor **ZVAR** witnesses that **t** is at the zeroth position in an environment of the form **t :: E**, whilst **SVAR** takes a proof that **t** is at some position *n* in the list **E** and produces a proof that **t** is at position *n* + 1 in **t' :: E**. (Note that the type of **ZVAR** builds in weakening.)

Now the typing rules for expressions are directly encoded in the types of the corresponding constructors of **Exp**. For example, the application constructor **APP** takes two expressions as argument, one of arrow type **ARR t1 t2** and the other of type **t1**, yielding an expression of type **t2**. The function constructor **LAM** takes an expression typed as **t2** under an environment extended with the argument of type **t1** and produces an expression of function type **ARR t1 t2**.

The inductive types `Var` and `Exp` are *indexed* by a (value) $t : \text{Ty}$ that is allowed to vary in the result types of the different constructors. In the functional programming community, datatypes whose (type) indices vary in this way are known as Generalized Algebraic Datatypes (*GADTs*). Indeed, a typed term representation similar to the above can be found in the standard test suite for the `ghc` compiler. For the `Exp` type, we have made the environment `E` into a *parameter*; note that this is scoped over all the constructors and remains the same in the return type of every constructor, though it does vary in the arguments to constructors, in a way that a functional programmer would call *non-regular* or *nested* (Bird and Meertens, 1998).

Definitions and statements involving strongly-typed terms are beautifully concise. For example, here is a complete definition of the “call-by-value” evaluation relation for closed expressions of type t :

```

Inductive Ev :  $\forall t, \text{Exp nil } t \rightarrow \text{Exp nil } t \rightarrow \text{Prop} :=
| \text{EvCONST} : \forall n, \text{Ev (CONST _ } n) (\text{CONST _ } n)
| \text{EvLAM} : \forall t1 t2 (e : \text{Exp [t1] } t2), \text{Ev (LAM e) (LAM e)}
| \text{EvSUCC} : \forall e n, \text{Ev e (CONST _ } n) \rightarrow \text{Ev (SUCC e) (CONST _ (n+1))}
| \text{EvPRED} : \forall e n, \text{Ev e (CONST _ } n) \rightarrow \text{Ev (PRED e) (CONST _ (n-1))}
| \text{EvIFZTHEN} : \forall t1 e (e1 e2 : \text{Exp nil } t1) v,
  \text{Ev e (CONST _ 0)} \rightarrow \text{Ev e1 v} \rightarrow \text{Ev (IFZ e e1 e2) v}
| \text{EvIFZELSE} : \forall t1 e (e1 e2 : \text{Exp nil } t1) v n,
  \text{Ev e (CONST _ (S n))} \rightarrow \text{Ev e2 v} \rightarrow \text{Ev (IFZ e e1 e2) v}
| \text{EvAPP} : \forall t1 t2 e v w (e1 : \text{Exp nil (ARR t1 t2)}) e2,
  \text{Ev e1 (LAM e)} \rightarrow \text{Ev e2 w} \rightarrow \text{Ev (STmExp \{ | w | \} e) v} \rightarrow
  \text{Ev (APP e1 e2) v}.$ 
```

Figure 2 presents essentially the same relation in more conventional form using named binders and type-free syntax. Again, the Coq encoding is a fairly direct translation of the conventional rules, but because we work with a strongly-typed representation, the fact that the definition of `Ev` typechecks in Coq gives us a proof that our evaluation relation preserves types almost for free, alongside its definition.²

Consider the most complex constructor `EvAPP`: it states that if an expression `e1` evaluates to a lambda expression `LAM e`, and `e2` evaluates to `w`, and `e` with its only free variable replaced by `w` evaluates to `v`, then `APP e1 e2` evaluates to `v`. However, we have presented our definitions out of order: neither the notation `\{ | w | \}`, which is intended to denote a substitution mapping the zero’th variable in the environment to `w`, nor the function `STmExp`, which applies a substitution to an expression, have yet been defined. How to define and work with strongly typed substitutions will be explained in the next couple of sections.

² Though one might reasonably object that the strongly-typed evaluation relation obscures the fact that evaluation does not depend on typing, of course.

$$\begin{array}{c}
\text{EvCONST} \frac{}{n \Downarrow n} \quad \text{EvLAM} \frac{}{\lambda x.e \Downarrow \lambda x.e} \quad \text{EvSUCC} \frac{e \Downarrow n}{\text{succ } e \Downarrow n+1} \\
\\
\text{EvPRED} \frac{e \Downarrow n}{\text{pred } e \Downarrow n-1} \quad \text{EvIFZTHEN} \frac{e \Downarrow 0 \quad e_1 \Downarrow v}{\text{ifz } e \ e_1 \ e_2 \Downarrow v} \\
\\
\text{EvIFZELSE} \frac{e \Downarrow n+1 \quad e_2 \Downarrow v}{\text{ifz } e \ e_1 \ e_2 \Downarrow v} \quad \text{EvAPP} \frac{e_1 \Downarrow \lambda x.e \quad e_2 \Downarrow v_2 \quad [x := v_2]e \Downarrow v}{e_1 \ e_2 \Downarrow v}
\end{array}$$

Fig. 2 Evaluation relation

3 Substitutions

We represent *typed substitutions* by functions that map variables typed in an environment E to expressions typed in an environment E' , written $\text{Sub } E \ E'$:

Definition $\text{Sub } E \ E' := \forall t, \text{Var } E \ t \rightarrow \text{Exp } E' \ t$.

The list-style notation $\{ | \ e_0 \ \dots, \ e_{n-1} \ | \}$ represents the substitution which takes variables numbered 0 to $n-1$ to expressions e_0 to e_{n-1} . The substitution is built up using the identity substitution idSub and an operator consSub which extends a substitution on the zero'th variable. Operations hdSub and tlSub can be used to decompose a substitution into the image of the zero'th variable and the remainder of the substitution.

Definition $\text{idSub } \{E\} : \text{Sub } E \ E := @\text{VAR } E$.

Program Definition $\text{consSub } \{E \ E' \ t\} (e : \text{Exp } E' \ t) (s : \text{Sub } E \ E') :$
 $\text{Sub } (t :: E) \ E' :=$
 $\text{fun } t' (v : \text{Var } (t :: E) \ t') \Rightarrow$
 $\text{match } v \text{ with}$
 $| \text{ZVAR } _ _ \Rightarrow e$
 $| \text{SVAR } _ _ _ v' \Rightarrow s _ _ v'$
 $\text{end}.$

Notation $\{ | \ e ; \dots ; f \ | \} := (\text{consSub } e \dots (\text{consSub } f \text{idSub}) \dots).$

Definition $\text{tlSub } \{E \ E' \ t\} (s : \text{Sub } (t :: E) \ E') : \text{Sub } E \ E' :=$
 $\text{fun } t' \ v \Rightarrow s \ t' (\text{SVAR } t \ v).$

Definition $\text{hdSub } \{E \ E' \ t\} (s : \text{Sub } (t :: E) \ E') : \text{Exp } E' \ t :=$
 $s \ t (\text{ZVAR } _ _).$

Notice here the use of Sozeau's **Program** tactic (Sozeau, 2007b), supporting GADT-style pattern matching in Coq. Looking at the first branch of the **match** in the definition of consSub , we see that the declared type of e is $\text{Exp } E' \ t$ whereas the type of the whole match expression is supposed to be $\text{Exp } E' \ t'$. But if $v : \text{Var } (t :: E) \ t'$ matches $\text{ZVAR } _ _$ we must have $t=t'$. The **Program** tactic generates and exploits this equation, producing a slightly

more complex CiC term ‘behind the scenes’. (McBride (2002) explains this transformation in more detail.)

Now let’s write the function that applies a substitution to an expression:

```

Fixpoint STmExp E E' t (s:Sub E E') (e:Exp E t) :=
  match e with
  | VAR _ v      ⇒ s _ v
  | CONST n      ⇒ CONST _ n
  | SUCC e       ⇒ SUCC (STmExp s e)
  | PRED e       ⇒ PRED (STmExp s e)
  | IFZ _ e e1 e2 ⇒ IFZ (STmExp s e) (STmExp s e1) (STmExp s e2)
  | APP _ _ e1 e2 ⇒ APP (STmExp s e1) (STmExp s e2)
  | LAM _ _ e    ⇒ LAM (STmExp (STmL s) e)
  end.

```

In the variable case we apply the substitution, and in most of the other cases we just do the obvious homomorphic thing. The interesting case is that for the LAM constructor, in which we have to apply the substitution under a binder. We want (indeed, the type system tells us we need) a function STmL that will *lift* the substitution to work over expressions in an extended environment. So let’s define STmL:

```

Program Definition
STmL {E E'} t (s:Sub E E') : Sub (t::E) (t::E') :=
  fun t' v ⇒
  match v with
  | ZVAR _ _ ⇒ VAR (ZVAR _ _)
  | SVAR _ _ v' ⇒ ShTmExp t (s _ v')
  end.

```

So far, so good, but we have now discovered that we have to be able to reinterpret (or ‘transport’) the expressions returned by our original substitution to work in the larger environment; hence we have postulated a *shift* operation ShTmExp with type $\text{Exp } E \ t' \rightarrow \text{Exp } (t::E) \ t'$. Intuitively, the shift just increments all the (term) variables in the expression. In Haskell with GADTs, we could implement this operation simply by applying a trivial substitution to the expression, using STmExp (**fun** t v ⇒ SVAR t' v). But this mutual recursion between substitution and shifting is not structurally recursive, and therefore is unacceptable in Coq.

One might now define ShTmExp directly, rather than in terms of substitutions. Because shifting itself has to be able to go under binders, one soon realizes that the type has to be generalized, but one can then easily define a shift function with type

$$\forall E E' \ t' \ t, \text{Exp } (E++E') \ t \rightarrow \text{Exp } (E++[t']++E') \ t$$

Unfortunately, working with that definition in Coq quickly becomes difficult. The problem is then properties of shifting have to be proved by induction over an argument expression of *arbitrary* type $\text{Exp } E \ t$. This involves recasting statements of the form

$\forall E E' t (e:\text{Exp } (E++E') t), \dots$

into the form

$\forall E0 t (e:\text{Exp } E0 t) E E', E0=E++E' \rightarrow \dots$

which requires passing in a proof of the equality. Sozeau's (2007a) formalization of the simply-typed lambda calculus uses this technique, but it requires the use of `eq_rect` cast operations and many lemmas that simply push them around in a manner that quickly becomes no fun at all.

4 Renamings

A better way of defining shifting is to observe (Adams, 2006; McBride, 2005) that a shift is an instance of a special, restricted kind of substitution: a *renaming*; that is, a map from variables to variables:

Definition $\text{Ren } E E' := \forall t, \text{Var } E t \rightarrow \text{Var } E' t.$

It is easy to define lifting for renamings, without running into issues with recursion:

```
Program Definition RTmL {E E' t}
  (r : Ren E E') : Ren (t::E) (t::E') := fun t' v =>
  match v with
  | ZVAR _ _ => ZVAR _ _
  | SVAR _ _ v' => SVAR _ (r _ v')
  end.
```

Applying a renaming to an expression is straightforward:

```
Fixpoint RTmExp E E' t (r:Ren E E') (e:Exp E t) :=
  match e with
  | VAR _ v      => VAR (r _ v)
  | CONST n      => CONST _ n
  | SUCC e       => SUCC (RTmExp r e)
  | PRED e       => PRED (RTmExp r e)
  | IFZ _ e e1 e2 => IFZ (RTmExp r e) (RTmExp r e1) (RTmExp r e2)
  | APP _ _ e1 e2 => APP (RTmExp r e1) (RTmExp r e2)
  | LAM _ _ e     => LAM (RTmExp (RTmL r) e)
  end.
```

And we can now define our shifting operation `ShTmExp` by applying a trivial renaming:

Definition $\text{ShTmExp } E t t' : \text{Exp } E t \rightarrow \text{Exp } (t'::E) t$
 $:= \text{RTmExp } (\text{fun } _ v \Rightarrow \text{SVAR } _ v).$

In order to make any use of these definitions, we have to prove a standard collection of lemmas about composing substitutions and so on. So as to fit the complete development of the theory of our simple language in the paper, and

because it is instructive in itself, we first define a couple of custom tactics for rewriting:

```

Ltac Rewrites E :=
  (intros; simpl; try rewrite E;
   repeat (match goal with | [H : context[eq _ _] |- _] => rewrite H end);
   auto).

Ltac ExtVar :=
  match goal with
  [ |- eq ?X ?Y ] =>
    (apply (@functional_extensionality_dep _ _ X Y) ;
     let t := fresh "t" in intro t;
     apply functional_extensionality;
     let v := fresh "v" in intro v;
     dependent destruction v; auto)
  end.

```

The `Rewrites` tactical applies the rewrite rule passed as argument, and then applies any equations that have been introduced as hypotheses, typically through induction. The `ExtVar` tactic applies extensionality for renamings or substitutions, introducing a type `t` and variable `v : Var _ t` into the context and then doing induction on `v` using the `dependent destruction` tactic, which (as with `Program`) takes care of generalizing the goal by the equalities generated by matching on `v`.

Using these tactics, we prove that lifting the identity gives the identity on the extended context, and hence that the action of the identity is the identity on terms:

Lemma `LiftIdSub` : $\forall E\ t, \text{STmL } (@\text{idSub } E) = @\text{idSub } (t::E)$.
Proof. `intros. ExtVar. Qed.`

Lemma `ActionIdSub` : $\forall E\ t\ (e : \text{Exp } E\ t), \text{STmExp idSub } e = e$.
Proof. `induction e; Rewrites LiftIdSub. Qed.`

The main downside of defining shift in terms of renaming is that we have defined everything twice: once for renaming, and once for substitution. And we now have *four* notions of composition:

```

Definition RcR {E E' E''} (r : Ren E' E'') r' :=
  (fun t v => r t (r' t v)) : Ren E E''.
Definition ScR {E E' E''} (s : Sub E' E'') r :=
  (fun t v => s t (r t v)) : Sub E E''.
Definition RcS {E E' E''} (r : Ren E' E'') s :=
  (fun t v => RTmExp r (s t v)) : Sub E E''.
Definition ScS {E E' E''} (s : Sub E' E'') s' :=
  (fun t v => STmExp s (s' t v)) : Sub E E''.

```

For each notion of composition we prove that lifting is preserved and that the action of a composition is a composition of actions. These lemmas must be proved *in order*, with each building on the previous:

Lemma LiftRcR : $\forall E E' E'' t (r:\text{Ren } E' E'') (r':\text{Ren } E E'),$
 $\text{RTmL } (t:=t) (\text{RcR } r r') = \text{RcR } (\text{RTmL } r) (\text{RTmL } r').$

Proof. intros. ExtVar. Qed.

Lemma ActionRcR : $\forall E t (e:\text{Exp } E t) E' E'' (r:\text{Ren } E' E'') (r':\text{Ren } E E'),$
 $\text{RTmExp } (\text{RcR } r r') e = \text{RTmExp } r (\text{RTmExp } r' e).$

Proof. induction e; Rewrites LiftRcR. Qed.

Lemma LiftScR : $\forall E E' E'' t (s:\text{Sub } E' E'') (r:\text{Ren } E E'),$
 $\text{STmL } (t:=t) (\text{ScR } s r) = \text{ScR } (\text{STmL } s) (\text{RTmL } r).$

Proof. intros. ExtVar. Qed.

Lemma ActionScR : $\forall E t (e:\text{Exp } E t) E' E'' (s:\text{Sub } E' E'') (r:\text{Ren } E E'),$
 $\text{STmExp } (\text{ScR } s r) e = \text{STmExp } s (\text{RTmExp } r e).$

Proof. induction e; Rewrites LiftScR. Qed.

Lemma LiftRcS : $\forall E E' E'' t (r:\text{Ren } E' E'') (s:\text{Sub } E E'),$
 $\text{STmL } (t:=t) (\text{RcS } r s) = \text{RcS } (\text{RTmL } r) (\text{STmL } s).$

Proof. intros. ExtVar. unfold RcS. simpl.

unfold ShTmExp. rewrite <- 2 ActionRcR. auto. Qed.

Lemma ActionRcS : $\forall E t (e:\text{Exp } E t) E' E'' (r:\text{Ren } E' E'') (s:\text{Sub } E E'),$
 $\text{STmExp } (\text{RcS } r s) e = \text{RTmExp } r (\text{STmExp } s e).$

Proof. induction e; Rewrites LiftRcS. Qed.

Lemma LiftScS : $\forall E E' E'' t (s:\text{Sub } E' E'') (s':\text{Sub } E E'),$
 $\text{STmL } (t:=t) (\text{ScS } s s') = \text{ScS } (\text{STmL } s) (\text{STmL } s').$

Proof. intros. ExtVar. simpl. unfold ScS. simpl.

unfold ShTmExp. rewrite <- ActionRcS. rewrite <- ActionScR. auto. Qed.

Lemma ActionScS : $\forall E t (e:\text{Exp } E t) E' E'' (s:\text{Sub } E' E'') (s':\text{Sub } E E'),$
 $\text{STmExp } (\text{ScS } s s') e = \text{STmExp } s (\text{STmExp } s' e).$

Proof. induction e; Rewrites LiftScS. Qed.

5 Example

Our experience with strong typing of terms is that the pain is worth it. Just as with typeful *programming*, typeful *proving* provides a framework for getting the definitions right and, generally speaking, the proofs then follow smoothly.

Type-indexed terms fit very well with typed-indexed semantics. To illustrate, we now describe how to give a set-theoretic denotational semantics

to our simple language. The formalization of a slightly more sophisticated domain-theoretic semantics for a language with recursion (Benton et al, 2009) follows just the same pattern, but working with a total language allows us here to focus on the type structure, without dragging in extraneous definitions concerning cpos and continuous functions.

We interpret NAT as Coq's `nat` type, ARR as Coq's \rightarrow , and environments as iterated product:

```
Fixpoint SemTy t :=
  match t with
  | NAT  $\Rightarrow$  nat
  | ARR t1 t2  $\Rightarrow$  SemTy t1  $\rightarrow$  SemTy t2
  end.
```

```
Fixpoint SemEnv E :=
  match E with
  | nil  $\Rightarrow$  unit
  | t :: E  $\Rightarrow$  prodT (SemTy t) (SemEnv E)
  end.
```

It is then straightforward to give a meaning to variables of type `Var E t` and (open) expressions of type `Exp E t` as functions of type `SemEnv E \rightarrow SemTy t`:

```
Fixpoint SemVar E t (v:Var E t) : SemEnv E  $\rightarrow$  SemTy t :=
  match v with
  | ZVAR _ _  $\Rightarrow$  fun se  $\Rightarrow$  fst se
  | SVAR _ _ _ v  $\Rightarrow$  fun se  $\Rightarrow$  SemVar v (snd se)
  end.
```

```
Require Import EqNat.
```

```
Fixpoint SemExp E t (e:Exp E t) : SemEnv E  $\rightarrow$  SemTy t :=
  match e with
  | VAR _ v  $\Rightarrow$  SemVar v
  | CONST n  $\Rightarrow$  fun se  $\Rightarrow$  n
  | SUCC e  $\Rightarrow$  fun se  $\Rightarrow$  SemExp e se + 1
  | PRED e  $\Rightarrow$  fun se  $\Rightarrow$  SemExp e se - 1
  | IFZ _ e e1 e2  $\Rightarrow$  fun se  $\Rightarrow$  if beq_nat (SemExp e se) 0
    then SemExp e1 se else SemExp e2 se
  | APP _ _ e1 e2  $\Rightarrow$  fun se  $\Rightarrow$  SemExp e1 se (SemExp e2 se)
  | LAM _ _ e  $\Rightarrow$  fun se  $\Rightarrow$  fun x  $\Rightarrow$  SemExp e (x,se)
  end.
```

Notice how natural these definitions are: the typed de Bruijn representation for variables fits perfectly with the use of pairing to extend environments.

In order to prove that the semantics is sound, we first need to prove a lemma showing that the semantics commutes with substitution. As with the syntactic proofs concerning composition, this lemma must be boot-strapped from an analogous lemma concerning renaming, as follows:

Lemma SemCommutesWithRen :

$$\forall E \ t \ (e : \text{Exp } E \ t) \ E' \ (r : \text{Ren } E \ E'),$$

$$\forall se, \text{SemExp } e \ (\text{SemRen } r \ se) = \text{SemExp } (\text{RTmExp } r \ e) \ se.$$

Lemma SemCommutesWithSub :

$$\forall E \ t \ (e : \text{Exp } E \ t) \ E' \ (s : \text{Sub } E \ E'),$$

$$\forall se, \text{SemExp } e \ (\text{SemSub } s \ se) = \text{SemExp } (\text{STmExp } s \ e) \ se.$$

We can now prove *soundness*: if an expression e evaluates to a value v then the denotation of e is the denotation of v . The above lemma is used in the crucial `EvAPP` case in the proof.

Theorem Soundness :

$$\forall t \ (e : \text{Exp } \text{nil } t) \ v, \text{Ev } e \ v \rightarrow \text{SemExp } e = \text{SemExp } v.$$

The semantics is also *adequate*: if the denotation of a closed expression e of base type is n , then e evaluates to `CONST _ n`. To prove adequacy, we use the standard method of defining a logical relation between syntax and semantics.

Definition evAndRel $R \ t \ (e : \text{Exp } \text{nil } t) \ (d : \text{SemTy } t) :=$

$$\exists v, \text{Ev } e \ v \wedge R \ t \ v \ d.$$

Fixpoint rel $t : \text{Exp } \text{nil } t \rightarrow \text{SemTy } t \rightarrow \text{Prop} :=$

match t **with**

- | NAT \Rightarrow **fun** $v \ n \Rightarrow v = \text{CONST } _ \ n$
- | ARR $t1 \ t2 \Rightarrow$ **fun** $v \ f \Rightarrow \exists e, v = \text{LAM } e$
 $\wedge \forall x \ w, @rel \ t1 \ w \ x \rightarrow$
 $@evAndRel \ rel \ t2 \ (\text{STmExp } \{|w|\} \ e) \ (f \ x)$

end.

This is extended to environments:

Fixpoint relEnv $E : \text{Sub } E \ \text{nil} \rightarrow \text{SemEnv } E \rightarrow \text{Prop} :=$

match E **with**

- | nil \Rightarrow **fun** $s \ se \Rightarrow \text{True}$
- | $t :: E \Rightarrow$ **fun** $s \ se \Rightarrow$
 $@rel \ t \ (\text{hdSub } s) \ (\text{fst } se) \wedge @relEnv \ E \ (t1Sub \ s) \ (\text{snd } se)$

end.

Then we can prove a fundamental theorem and adequacy is a corollary:

Theorem FundamentalTheorem :

$$\forall E \ t \ e \ se \ s, @relEnv \ E \ s \ se \rightarrow$$

$$@evAndRel \ rel \ t \ (\text{STmExp } s \ e) \ (\text{SemExp } e \ se).$$

Corollary Adequacy :

$$\forall (e : \text{Exp } \text{nil } \text{NAT}) \ n, \text{SemExp } e \ tt = n \rightarrow \text{Ev } e \ (\text{CONST } _ \ n).$$

The whole development, including syntax and semantics, is roughly 200 lines of definition and 120 lines of proof.

6 Extensions

Our approach extends easily to more complex uses of variable binding. For example, here is a constructor for recursive functions, which might be written $\text{rec } f(x : t_1) : t_2 = e$ in more conventional notation.

```
Inductive Exp E : Ty → Type :=
  ...
| REC : ∀ t1 t2, Exp (t1::ARR t1 t2::E) t2 → Exp E (ARR t1 t2).
```

It's also easy to model SML-style pattern matching. Assuming type constructors **PROD** and **SUM** for products and sums, we can define SML-style pattern expressions of type $\text{Pat } E \ t$ where t is the type of the whole pattern and E lists the types of variables mentioned in the pattern, reading from left to right.

```
Inductive Pat : Env → Ty → Type :=
| PPAIR : ∀ E1 E2 t1 t2, Pat E1 t1 → Pat E2 t2 →
    Pat (E1++E2) (PROD t1 t2)
| PVAR : ∀ t, Pat [t] t
| PWILD : ∀ t, Pat [] t
| PFAIL : ∀ t, Pat [] t
| PAS : ∀ E t, Pat E t → Pat (t::E) t
| PINL : ∀ E t1 t2, Pat E t1 → Pat E (SUM t1 t2)
| PINR : ∀ E t1 t2, Pat E t2 → Pat E (SUM t1 t2).
```

We can then use pattern expressions in a ‘let’ construct, as follows:

```
Inductive Exp E : Ty → Type :=
  ...
| LETPAT : ∀ E' t1 t2, Exp E t1 → Pat E' t1 → Exp (E'++E) t2 →
    Exp E t2.
```

7 Abstracting maps

For languages larger than **Exp**, the necessity to do so many things twice, once for renamings, and once for substitutions, becomes somewhat painful. At least some cutting-and-pasting of definitions and proofs can be avoided by observing the commonality between renaming and substitution, abstracting both notions into a single **Map** type that is parameterized on the type constructor used to construct its target, namely **Var** (for renamings) or **Exp** (for substitutions) (McBride, 2005).

```
Definition Map (P:Env → Ty → Type) E E' :=
  ∀ t, Var E t → P E' t.
Definition Ren := Map Var.
Definition Sub := Map Exp.
```

We then package up the fundamental operations used in lifting and the action of a map on an expression:

```

Record MapOps (P : Env → Ty → Type) :=
{
  vr : ∀ E t, Var E t → P E t;
  vl : ∀ E t, P E t → Exp E t;
  wk : ∀ E t t', P E t → P (t' :: E) t
}.

```

Here `vr` is the embedding of a variable into the target type, namely the identity for renamings, and the `Var` constructor for expressions. The `vl` is an embedding into `Exp`, namely the `Var` constructor for variables, and the identity for expressions. Finally `wk` is the operation that maps into a weaker context, namely `SVAR` for variables, and `ShTmExp` (shift) for expressions. We can then define ‘generic’ lifting and application functions, parameterized by `P` and a package of type `MapOps P`.

```

Program Definition MTmL P (ops : MapOps P) E E' t (m:Map P E E')
: Map P (t::E) (t::E') :=
  fun t' => fun v => match v with
  | ZVAR _ _ => vr ops (ZVAR _ _)
  | SVAR _ _ v' => wk ops _ _ (m _ v')
end.

```

```

Fixpoint MTmExp P (ops : MapOps P) E E' t (m:Map P E E') (e:Exp E t) :=
match e with
| VAR _ v => vl ops _ _ (m _ v)
| CONST n => CONST _ n
| SUCC e => SUCC (MTmExp ops m e)
| PRED e => PRED (MTmExp ops m e)
| IFZ _ e1 e2 => IFZ (MTmExp ops m e) (MTmExp ops m e1)
               (MTmExp ops m e2)
| LAM _ _ e => LAM (MTmExp ops (MTmL ops m) e)
| APP _ _ e1 e2 => APP (MTmExp ops m e1) (MTmExp ops m e2)
end.

```

The instantiation to give operations on renamings and substitutions is straightforward:

```

Definition RMapOps := Build_MapOps Var (fun _ _ v => v) VAR (@SVAR).
Definition RTmExp := MTmExp RMapOps.
Definition RTmL := MTmL RMapOps.

```

```

Definition ShTmExp E t t' : Exp E t → Exp (t'::E) t
:= RTmExp (fun _ v => SVAR _ v).
Definition SMapOps := Build_MapOps Exp VAR (fun _ _ v => v) ShTmExp.
Definition STmExp := MTmExp SMapOps.
Definition STmL := MTmL SMapOps.

```

The composition lemmas one then proves are essentially the same as before, and these are easily dispatched tactically; the main advantage of the map

abstraction is in avoiding repeated definitions. One might also usefully parameterize `MTmExp` by a monad, to support generic effectful traversals.

8 Polymorphism

We now turn to applying the same basic ideas to an intrinsic encoding of the second order polymorphic lambda calculus, System F. We will use the renamings and substitutions idea for both types and terms. There is a mild combinatorial explosion in the number of forms of application and composition (e.g. the action of a type substitution on a term renaming) but, fortunately, not *all* combinations show up in establishing the lemmas that clients need.

Types now contain type variables, represented again by de Bruijn indices. A type variable context is represented simply by its length, a natural number `u` saying how many type variables are available. We then define Coq types for well-formed type variables and types in context:

```
Inductive TyVar : nat → Type :=
| ZTYVAR : ∀ u, TyVar (S u)
| STYVAR : ∀ u, TyVar u → TyVar (S u).
```

```
Inductive Ty u : Type :=
| TYVAR : TyVar u → Ty u
| ARR : Ty u → Ty u → Ty u
| ALL : Ty (S u) → Ty u.
```

Type renamings and substitutions are defined as follows:

```
Definition RenT u w := TyVar u → TyVar w.
```

```
Definition SubT u w := TyVar u → Ty w.
```

Lifting of renamings and the action of a renaming on a type are given by

```
Program Definition RTyL u w (r:RenT u w) : RenT (S u) (S w) :=
fun var ⇒
  match var with
  | ZTYVAR _ ⇒ (ZTYVAR _)
  | STYVAR _ var' ⇒ STYVAR (r var')
end.
```

```
Fixpoint RTyT u w (r:RenT u w) (t:Ty u) : Ty w :=
  match t with
  | TYVAR v ⇒ TYVAR (r v)
  | ARR t1 t2 ⇒ ARR (RTyT r t1) (RTyT r t2)
  | ALL t ⇒ ALL (RTyT (RTyL r) t)
end.
```

and again, shifting is defined as a special renaming, which is in turn used to define the action of a substitution on a type:

Definition $\text{ShTyT } u : \text{Ty } u \rightarrow \text{Ty } (S \ u) := \text{RTyT } (@\text{STYVAR } _)$.

Program Definition $\text{STyL } u \ w \ (s:\text{SubT } u \ w) : \text{SubT } (S \ u) \ (S \ w) :=$
 $\text{fun } v \Rightarrow$
 $\text{match } v \text{ with}$
 $| \text{ZTYVAR } _ \Rightarrow \text{TYVAR } (\text{ZTYVAR } _)$
 $| \text{STYVAR } _ \ v' \Rightarrow \text{ShTyT } (s \ v')$
 end.

Fixpoint $\text{STyT } u \ w \ (s:\text{SubT } u \ w) \ (t:\text{Ty } u) : \text{Ty } w :=$
 $\text{match } t \text{ with}$
 $| \text{TYVAR } v \Rightarrow s \ v$
 $| \text{ARR } t1 \ t2 \Rightarrow \text{ARR } (\text{STyT } s \ t1) \ (\text{STyT } s \ t2)$
 $| \text{ALL } t \Rightarrow \text{ALL } (\text{STyT } (\text{STyL } s) \ t)$
 end.

Following the pattern we used earlier for simply typed terms, we now define notations for particular type substitutions (we use $[| \ t1, \dots, tn \ |]$), composition of type renamings and type substitutions, and prove appropriate lemmas.

We are now ready to introduce strongly polymorphically typed *terms*. A well-formed term variable environment in a type variable context with u free variables is represented, again using de Bruijn indices, as a list of u -types, and the action of type substitutions on term environments is just given by mapping:

Definition $\text{Env } u := \text{list } (\text{Ty } u)$.

Fixpoint $\text{STyE } u \ w \ (\text{sub}: \text{SubT } u \ w) \ (\text{env}: \text{Env } u) : \text{Env } w :=$
 $\text{match } \text{env} \text{ with}$
 $| \text{nil} \Rightarrow \text{nil}$
 $| T::TS \Rightarrow \text{STyT } \text{sub } T :: \text{STyE } \text{sub } TS$
 end.

Next we define a type for typed variables in a given term and type variable context:

Inductive $\text{Var } u : \text{Env } u \rightarrow \text{Ty } u \rightarrow \text{Type} :=$
 $| \text{ZVAR} : \forall \text{env } ty, \text{Var } (ty :: \text{env}) \ ty$
 $| \text{SVAR} : \forall \text{env } ty' \ ty, \text{Var } \text{env } ty \rightarrow \text{Var } (ty' :: \text{env}) \ ty.$

For convenience, we re-express type shifting as a substitution:

Definition $\text{shSubT } u : \text{SubT } u \ (S \ u) := \text{fun } v \Rightarrow \text{TYVAR } (\text{STYVAR } v).$
Implicit Arguments $\text{shSubT } []$.

and the definition of terms is then a pleasingly direct translation of the ‘normal’ typing rules for the polymorphic lambda calculus:

Inductive $\text{Exp } u \ (E:\text{Env } u) : \text{Ty } u \rightarrow \text{Type} :=$

```

| VAR : ∀ t, Var E t → Exp E t
| LAM : ∀ t1 t2, Exp (t1 :: E) t2 → Exp E (ARR t1 t2)
| APP : ∀ t1 t2, Exp E (ARR t1 t2) → Exp E t1 → Exp E t2
| TAPP : ∀ t, Exp E (ALL t) → ∀ t':Ty u, Exp E (STyT [| t' |] t)
| TABS : ∀ t, Exp (u:=S u) (STyE (shSubT _) E) t → Exp E (ALL t).

```

Again, there are no proofs of equalities passed as arguments to any of the constructors; everything is built up by inductive definitions. Note how type substitution shows up in the type of the `TAPP` constructor, and how the environment is shifted in the type of the argument to `TABS`.

The broad pattern of the formalization of operations on the polymorphic language follows that of the simply-typed case. We have type renamings and type substitutions, and term renamings and term substitutions. We choose to abstract the two kinds of traversals over terms into a more general notion of mapping, along the lines described in Section 7, with an extra operation component to account for the action of type substitutions.

Working with the polymorphic encoding is qualitatively more tricky than was the case for the simply-typed language, however. Although the definitions of polymorphic types and terms are simple and elegant, we have this time *not* managed to avoid explicit uses of type equalities when we come to define functions working over that syntax. Here, for example, is the definition of the action of type substitutions on terms:

```

Fixpoint STyExp u w (s:SubT u w) (E:Env u) t (e:Exp E t)
: Exp (STyE s E) (STyT s t) :=
  match e with
  | VAR _ v ⇒ VAR (STyVar s v)
  | APP _ _ e1 e2 ⇒ APP (STyExp s e1) (STyExp s e2)
  | LAM _ _ e ⇒ LAM (STyExp s e)
  | TAPP _ e t' ⇒ cast (STyExp_cast1 _ _ _ _)
                      (TAPP (STyExp s e) (STyT s t'))
  | TABS _ e ⇒ TABS (cast (STyExp_cast2 _ _ _ _)
                      (STyExp (STyL s) e))
end.

```

Observe that we have been forced to make explicit applications of `cast`, the obvious operation of type $\forall A B : \text{Type}, A = B \rightarrow A \rightarrow B$, to make this definition typecheck. The two type equalities passed to `cast` are

```

Lemma STyExp_cast1 : ∀ u w (sub: SubT u w) (env: Env u)
                      (ty : Ty (S u)) (ty' : Ty u),
  @eq Type
  (Exp (STyE sub env) (STyT [| STyT sub ty' |] (STyT (STyL sub) ty)))
  (Exp (STyE sub env) (STyT sub (STyT [| ty' |] ty))).

```

```

Lemma STyExp_cast2 : ∀ u w (sub : SubT u w) (env: Env u) (ty: Ty (S u)),
  @eq Type
  (Exp (STyE (STyL sub) (STyE (shSubT u) env)) (STyT (STyL sub) ty))

```

(Exp (STyE (shSubT _) (STyE sub env)) (STyT (STyL sub) ty)).

which establish (with a one-line proof in each case) that the inferred and declared types are actually equal in the **STyExp** clauses for type application and type abstraction, respectively.

Where Coq definitions can be easily rephrased in a way that avoids the necessity to do this kind of casting, they usually should be. When they cannot, however, we have sometimes found ourselves overwhelmed by the complexities of trying to do more-or-less *ad hoc*, on the fly dependent rewrites with various forms of proof irrelevance (be they axiomatic or proved). Our formalization of polymorphic lambda calculus makes more disciplined and stylised use of the weapon of heterogeneous equality (McBride, 2002), which turns out to be more effective than aimless slashing.

We prove lemmas that all our definitions are congruences with respect to **JMeq** in their non-trivially dependent arguments, and Leibniz equality in the others. These lemmas are tedious to state but essentially just boilerplate: some simple custom tactics prove them immediately, and there seems no reason why they shouldn't be automatically generated. Here's an example:

Lemma APP_JMcong: $\forall u \text{ (env env': Env } u) \text{ ty1 ty2 ty1' ty2'}$
 $(v1: \text{Exp env (ARR ty1 ty2)}) (v2: \text{Exp env ty1})$
 $(v1': \text{Exp env' (ARR ty1' ty2')}) (v2': \text{Exp env' ty1'}),$
 $\text{JMeq } v1 \ v1' \rightarrow \text{JMeq } v2 \ v2' \rightarrow \text{ty1} = \text{ty1}' \rightarrow \text{ty2} = \text{ty2}' \rightarrow \text{env} = \text{env'}$
 $\rightarrow \text{JMeq (APP } v1 \ v2) \text{ (APP } v1' \ v2').$

Proof. intros. JMsubst. reflexivity. **Qed.**

Now those standard lemmas about about renamings and substitutions that would not otherwise typecheck can be expressed using **JMeq**. For example, the lemma that the action on expressions of the composition of two type substitutions is the composition of the actions looks like this:

Lemma STyExp_ss: $\forall u \text{ env ty (exp: Exp env ty) } v \ w$
 $(\text{sub2: SubT } v \ w) (\text{sub1: SubT } u \ v),$
 $\text{JMeq (STyExp sub2 (STyExp sub1 exp))}$
 $(\text{STyExp (sub2 @ss@ sub1) exp}).$

where **@ss@** is notation of composition of type substitutions. The proofs of these lemmas are straightforward inductions, essentially just as before, except for the application of the appropriate (boilerplate) congruence property for each constructor. In the cases where the type of the constructor involves a cast, however, we can now easily 'absorb' the specific cast appearing in the type into the more uniform and generic **JMeq** judgement that we are trying to prove using some simple lemmas such as the following, which removes a cast from the left hand side of the goal:

Lemma cast_elim_cong : $\forall (A \ B \ C: \text{Type}) \text{ (pf: } A = C) \text{ (a:A) (b:B)},$
 $\text{JMeq } a \ b \rightarrow \text{JMeq (cast pf } a) \ b.$

Some of the lemmas whose proofs make use of heterogeneous equality internally only mention Leibniz equality in their statements, but this is not the

case for all the lemmas that one needs in applications. So the uses of `JMeq` do ‘leak out’ of the syntax module into clients, rather than being encapsulated as one might hope. In our work on compiler correctness for a polymorphic language (Benton and Hur, 2009b), for example, the definitions of logical relations between high-level and low-level programs do explicitly involve heterogeneous equality. However, working with `JMeq` uniformly, and from the start, does seem to us to work well, especially compared to pushing particular type equalities around. The basic definitions and lemmas concerning types, terms, renamings and substitutions for the polymorphic language come in at around 1000 lines of Coq, which does not seem *completely* unreasonable.

9 Discussion

We have explained how to define and work with strongly-typed term representations of both simply-typed and polymorphic languages in Coq. The key ideas include the bootstrapping of definitions and lemmas about substitutions in terms of their counterparts for the simpler notion of renamings, and the uniform use of heterogeneous equality in the case of quantified types.

We have used intrinsically typed representations like these in formalizing and proving some non-trivial results about the semantics and compilation of typed languages. Our experience has been that the initial complexity over an extrinsic representation really does pay off - one gets all the stuff to do with the static type system out of the way in the beginning and then when it comes to doing interesting things with those terms, the type system becomes a useful form of scaffolding, with the metalanguage type checker helping ensure that definitions and lemmas make sense, rather than a constant nagging extra obligation.

It is possible to work with encodings that are ‘partially’ intrinsic. Some researchers have used syntax definitions that are well-scoped (i.e. whose types express an upper bound on their free de Bruijn indices) by definition, but are still actually typed by an extrinsic typing relation (Hirschowitz and Maggesi, 2009; Bird and Paterson, 1999; Adams, 2006). This seems particularly natural if one is trying to formalize type theory within type theory, as dependency naturally leads one to have to work with ‘pre-terms’ in the first instance.

There are a number of possible variations on the techniques we have used here. One possibility is to re-examine the ‘Haskell-style’ definition of shifting in terms of substitution. Although this is not structurally recursive, one *could* define it in Coq using well-founded induction, but we have not yet investigated how easy such a definition would be to work with. One can also represent intrinsically typed syntax in Coq using various kinds of higher-order abstract syntax. This style is already common in Twelf, and Chlipala (2008), for example, has done similar things in Coq.

References

- Adams R (2006) Formalized metatheory with terms represented by an indexed family of types. In: *Types for Proofs and Programs*, LNCS, vol 3839
- Altenkirch T, Reus B (1999) Monadic presentations of lambda terms using generalized inductive types. In: *Computer Science Logic*, LNCS, vol 1683
- Benton N, Hur CK (2009a) Biorthogonality, step-indexing and compiler correctness. In: *Proceedings of the 14th ACM SIGPLAN International Conference on Functional Programming (ICFP)*
- Benton N, Hur CK (2009b) Compositional compiler correctness with quantified types, draft
- Benton N, Kennedy AJ, Varming C (2009) Some domain theory and denotational semantics in Coq. In: *Theorem Proving in Higher Order Logics (TPHOLs)*, LNCS, vol 5674
- Bird R, Meertens L (1998) Nested datatypes. In: *Proceedings of the 4th International Conference on Mathematics of Program Construction (MPC)*, LNCS, vol 1422
- Bird R, Paterson R (1999) de Bruijn notation as a nested datatype. *Journal of Functional Programming* 9:77–91
- Cheney J, Hinze R (2003) First-class phantom types. Tech. Rep. TR2003-1901, Cornell University
- Chlipala A (2008) Parametric higher-order abstract syntax for mechanized semantics. In: *Proceedings of the 13th ACM SIGPLAN International Conference on Functional Programming (ICFP)*
- Coquand C (2002) A formalized proof of the soundness and completeness of a simply typed lambda-calculus with explicit substitutions. *Higher-Order and Symbolic Computation* 15:57–90
- Danielsson NA (2007) A formalization of a dependently typed language as an inductive-recursive family. In: *Types for Proofs and Programs*, LNCS, vol 4502
- Goguen H, McKinna J (1997) Candidates for substitution. Tech. Rep. ECS-LFCS-97-358, University of Edinburgh
- Harper R, Honsell F, Plotkin G (1993) A framework for defining logics. *Journal of the ACM* 40(1):143–184
- Hirschowitz A, Maggesi M (2009) Nested abstract syntax in Coq, draft
- Kennedy A, Russo C (2005) Generalized algebraic datatypes and object oriented programming. In: *ACM Conference on Object Oriented Programming Systems, Languages and Applications (OOPSLA)*
- McBride C (2002) Elimination with a motive. In: *Types for Proofs and Programs*, LNCS, vol 2002
- McBride C (2005) Type-preserving renaming and substitution, draft
- McBride C, McKinna J (2004) The view from the left. *Journal of Functional Programming* 14(1):69–111
- Sheard T (2004) Languages of the future. In: *ACM Conference on Object Oriented Programming Systems, Languages and Applications (OOPSLA)*

-
- Sozeau M (2007a) A dependently-typed formalization of simply-typed lambda-calculus: substitution, denotation, normalization, draft
- Sozeau M (2007b) PROGRAM-ing finger trees in COQ. In: Proceedings of the 12th ACM SIGPLAN International Conference on Functional Programming (ICFP)
- Xi H, Chen C, Chen G (2003) Guarded recursive datatype constructors. In: Proceedings of the 30th ACM SIGPLAN-SIGACT Symposium on Principles of Programming Languages (POPL)